clear all
set more off
set matsize 10000
tempfile ubigeo temp


local lista trust_police trust_army trust_church trust_newsp trust_rtv trust_parties trust_regi trust_jury trust_parliament trust_jne trust_onpe
local controls schooling male rural native married head 

use mergedp, clear

egen voted_bn = rowmax(voted_blank voted_null)
recode voted_bn voted_blank voted_null (. = 0) if voted == 0

** LABELS
label var todoA "Army"
label var todoT "Terrorist"

* keep only violence fully explained by either A or T, or none
gen rateA = todoA / todoAT
gen rateT = todoT / todoAT

drop if rateA < 1 & rateA > 0
drop if rateT < 1 & rateT > 0

******* Results
drop if vote_matter == .
local lista vote_matter voted voted_blank voted_null voted_bn partN dem_works dem_matters1 gov_demo1 gov_auto1

local r = 1
foreach var in `lista' {
	replace `var' = `var' * 100 if `r' != 6
	reghdfe `var' todoA todoT `controls', abs(ubigeo1993 ubigeonac1993 i.ubigeo1993#i.ubigeonac1993 i.age i.year i.provid#c.trend) vce(cluster ubigeonac1993) keepsin
	est store reg`r'
	sum `var' if e(sample) == 1
	local y = r(mean)
	sum todoA if e(sample) == 1 & todoA
	local sda = r(mean)
	sum todoT if e(sample) == 1 & todoT
	local sdt = r(mean)
	local na`r' = (round(((_b[todoA]) * `sda') / `y', 0.0001)) * 100
	local nt`r' = (round(((_b[todoT]) * `sdt') / `y', 0.0001)) * 100
	local ++r
}

noisily estout reg* ///
using Table_11_0.tex, style(tex) c(b(fmt(2) star) se(par fmt(2))) ///
	label collabels(, none) eqlabels(none) mlabels(, none) varlabel(_cons " $ Control $ ") starlevel(* .1 ** .05 *** .01) ///
	keep(todoA todoT) ///
	order(todoA todoT) ///
	stats(r2 N N_clust, fmt(%9.2f %9.0g) labels(`"$ R ^{2}$"' "Observations" "Clusters")) replace 

texdoc init Table_11.tex, replace
tex \begin{table}[h]
tex \caption{Impact of armed conflict on voting, participation and democratic beliefs: sample of individuals that experienced wither Army or Terrorist violence}
tex \label{tab:Table_11}
tex \begin{center}
tex \resizebox{16cm}{!}{
tex \begin{tabular}{lcccccccccc} \hline \hline
tex \\
tex & Voting & \multicolumn{4}{c}{Voted in last elections}& Participation & \multicolumn{2}{c}{Democracy} & \multicolumn{2}{c}{Type of preferred gov.} \\
tex & matters & Yes & ...blank & ...null & ...blank or & organisations & works & matters & Democracy & Autocracy \\
tex &&&&&Null&&&&& \\
tex &(1) &(2) &(3) &(4) &(5) &(6) &(7) &(8) &(9) &(10) \\
tex \hline
tex \input{Table_11_0.tex}
tex &&&&&&&&& \\
tex \multicolumn{8}{l}{\textbf{Effect of 1 s.d. change in conflict measure as \% of mean of dependent variable:}} \\
tex Army & `na1' & `na2' & `na3' & `na4' & `na5' & `na6' & `na7' & `na8' & `na9' & `na10' \\
tex Terrorist & `nt1' & `nt2' & `nt3' & `nt4' & `nt5' & `nt6' & `nt7' & `nt8' & `nt9' & `nt10' \\
tex &&&&&&&&& \\
tex \hline
tex \end{tabular}}
tex \end{center}
tex \end{table}	
